Solving rolling disk inside another using symbolic computation

by Nasser M. Abbasi. April 20, 2011

In[1]:=
```
Needs["Notation`"]
```

In[2]:=
```
Symbolize[m1]
Symbolize[m2]
Symbolize[I1]
Symbolize[I2]
```

Solve for ω to meet the no-slip condition

In[9]:=
```
Clear["Global`*"]
noSlipEquation = (R - r) θ'[t] == R ϕ'[t] + r ω;
ω = ω /. First@Solve[noSlipEquation, ω]
```

Out[11]=

$$\frac{-r\,\theta'[t] + R\,\theta'[t] - R\,\phi'[t]}{r}$$

Find T and V and find the Lagrangian

In[33]:=

$$T = \frac{1}{2} m_1 (R\,\phi'[t])^2 + \frac{1}{2} I_1 \phi'[t]^2 + \frac{1}{2} m_2 \left((R\,\phi'[t] - (R - r)\,\theta'[t]\,\mathrm{Cos}[\theta[t]])^2 + ((R - r)\,\theta'[t]\,\mathrm{Sin}[\theta[t]])^2\right) + \frac{1}{2} I_2 \omega^2$$

Out[33]=

$$\frac{1}{2} I_1 \phi'[t]^2 + \frac{1}{2} m_1 R^2 \phi'[t]^2 + \frac{I_2 (-r\,\theta'[t] + R\,\theta'[t] - R\,\phi'[t])^2}{2 r^2} + \frac{1}{2} m_2 \left((-r + R)^2 \mathrm{Sin}[\theta[t]]^2 \theta'[t]^2 + (-((-r + R)\,\mathrm{Cos}[\theta[t]]\,\theta'[t]) + R\,\phi'[t])^2\right)$$

In[34]:=

$$V = -m_2\, g\, (R - r)\, \mathrm{Cos}[\theta[t]]$$

Out[34]=

$$-g\, m_2 (-r + R)\, \mathrm{Cos}[\theta[t]]$$

In[35]:=
```
L = (T - V) // FullSimplify
```

Out[35]=

$$g\, m_2 (-r + R)\, \mathrm{Cos}[\theta[t]] + \frac{1}{2 r^2}\left(\left(I_2 + m_2 r^2\right)(r - R)^2 \theta'[t]^2 + 2 (r - R) R \left(I_2 + m_2 r^2 \mathrm{Cos}[\theta[t]]\right) \theta'[t]\,\phi'[t] + \left(I_1 r^2 + \left(I_2 + (m_1 + m_2) r^2\right) R^2\right) \phi'[t]^2\right)$$

Solve for ϕ''[t]. Note the generalized force is zero

In[36]:=
```
equationOfMotion1 = D[D[L, ϕ'[t]], t] - D[L, ϕ[t]] == 0 // Simplify
```

Out[36]=
$$\frac{1}{r^2}\left(-m_2 r^2 (r-R) R \operatorname{Sin}[\theta[t]]\, \theta'[t]^2 + (r-R) R \left(I_2 + m_2 r^2 \operatorname{Cos}[\theta[t]]\right) \theta''[t] + \left(I_1 r^2 + \left(I_2 + (m_1+m_2) r^2\right) R^2\right) \phi''[t]\right) == 0$$

Solve for θ''[t] . Note the generalized force is zero

In[37]:=
```
equationOfMotion2 = D[D[L, θ'[t]], t] - D[L, θ[t]] == 0 // Simplify
```

Out[37]=
$$\frac{(r-R)\left(-g\, m_2 r^2 \operatorname{Sin}[\theta[t]] + \left(I_2 + m_2 r^2\right)(r-R)\, \theta''[t] + R\left(I_2 + m_2 r^2 \operatorname{Cos}[\theta[t]]\right) \phi''[t]\right)}{r} == 0$$

Define problem parameters

In[38]:=
```
parms = {g -> 9.81, R -> 1, r -> 0.1, m1 -> 10, m2 -> 1};
parms = Union[parms, {I1 -> m1 R^2/2, I2 -> m2 r^2/2} /. parms]
```

Out[39]= $\{g \to 9.81,\ I_1 \to 5,\ I_2 \to 0.005,\ m_1 \to 10,\ m_2 \to 1,\ r \to 0.1,\ R \to 1\}$

Numerically solve the equations of motion using some initial conditions

In[40]:=
```
s = NDSolve[{equationOfMotion1, equationOfMotion2, ϕ[0] == 30 Degree,
      ϕ'[0] == -2, θ[0] == 0, θ'[0] == -2} /. parms, {ϕ[t], θ[t]}, {t, 0, 30}]
```

Out[40]=
```
{{ϕ[t] → InterpolatingFunction[ Domain: {{0., 30.}} Output: scalar ][t],
  θ[t] → InterpolatingFunction[ Domain: {{0., 30.}} Output: scalar ][t]}}
```

Plot the solution, motion of small disk. We see the small disk makes an oscillatio9n motion inside the large disk as the large disk is rotating

In[42]:=
```
Plot[Evaluate[θ[t] /. s], {t, 0, 30}, PlotRange → All, Frame -> True,
  FrameLabel → {{"θ[t]", None}, {"time (sec)", "The solution θ[t] of small disk"}},
  FrameTicks → {Automatic, {-30 Degree, -15 Degree, 15 Degree, 30 Degree}}]
```

Out[42]=